Eric Doviak
07 March 2014

Quantitative Analysis
Problem #1 -- maximization of expected value of portfolio

A mortgage lender seeks to maximize the expected value of its portfolio. No generality is lost by examining the case of one loan.

E[port] = (1−p)*B + p*(V-L)

Lenders may not recover more than the principal balance through the foreclosure process. When the borrower is "underwater," the lender recovers less than the principal balance:

V-L < B

Probability of foreclosure is an increasing function of the "balance-to-value" ratio, (B/V):

p == p(B/V)
p' > 0

Assume that the sale value of the property and legal fees are exogenous to the lender. In other words, assume that V and L are exogenous.

In this simple model, the lender maximizes the expected value of the portfolio by choosing a principal balance for the borrower.

At first glance, this model may seem silly. After all, there are good legal reasons why a lender may not increase the borrower's principal balance after it has signed a mortgage agreement with the borrower.

Nothing prevents a lender from reducing the principal balance however. In cases where the borrower is "underwater," the borrower has a financial incentive to let the home go into foreclosure.
("Why pay another $200,000 for a home that is only worth $100,000?")

In such cases, the lender may have a financial incentive to reduce the borrower's principal balance.

Since we're working with probability, we need to specify a distribution. It's not necessary for the model, but it is necessary for this notebook. We'll use lognormal.

- capital P for the probability (the cumulative distribution)
- lowercase p for its derivative (the probability density function)

```
(%i1) /*  define the functions -- even though "L" is exogenous we will */
      /*  include it as variable, so that we can perform comparative   */
      /*  static exercises with "L" later                              */

      EPort(B,L) := B*(1-P(B/V)) + P(B/V)*(V-L)$
      dEP(B,L) := ''(diff(EPort(B,L),B))$

      print("")$
      print("expected value of portfolio")$
      print("")$
      print("EPort(B) = ",EPort(B,L))$
      print("")$
      print("")$
      print("expected value maximized when:")$
      print("")$
      print(("d EPort")/("d B")," = ",dEP(B,L)," = 0")$
      print("")$
```

expected value of portfolio

$$EPort(B) = P\left(\frac{B}{V}\right)(V-L)+B\left(1-P\left(\frac{B}{V}\right)\right)$$

expected value maximized when:

$$\frac{d\ EPort}{d\ B} = \left(\frac{d}{d\,B}P\left(\frac{B}{V}\right)\right)(V-L)-P\left(\frac{B}{V}\right)-B\left(\frac{d}{d\,B}P\left(\frac{B}{V}\right)\right)+1 = 0$$

Removing the functional notation simplifies those expressions.

```
(%i13) ep_simple  : subst(    P     ,  P(B/V)          , EPort(B,L))$
       dep_stepA  : subst("(dP/dB)", 'diff(P(B/V),B,1), dEP(B,L)  )$
       dep_simple : subst(    P     ,  P(B/V)          , dep_stepA )$

       print("")$
       print("expected value of portfolio")$
       print("")$
       print("EPort(B) = ",ep_simple)$
       print("")$
       print("")$
       print("expected value maximized when:")$
       print("")$
       print(("d EPort")/("d B")," = ",dep_simple," = 0")$
       print("")$
```

expected value of portfolio

$$EPort(B) = P\left(V-L\right)+B\left(1-P\right)$$

expected value maximized when:

$$\frac{d\ EPort}{d\ B} = (dP/dB)\left(V-L\right)-P-(dP/dB)\,B+1 = 0$$

Rearranging terms gives us Marginal Cost and Marginal Benefit.

```
   *  MC = 1-P
   *  MB = (dp/dB)*(B-V+L)

By the chain rule:

   d P(B/V)       d P(B/V)   d (B/V)       1   d P(B/V)      p
  ----------  =  ----------*---------  =  ---*----------  =  ---
     d B          d (B/V)     d B          V    d (B/V)       V
```

```
(%i26) MB(B,L):= ((B-V+L)/V) * p(B/V) $
       MC(B,L):= 1 - P(B/V) $

       print("")$
       print("marginal benefit of reducing principal balance")$
       print("")$
       print("MB(B) = ",MB(B,L),"")$
       print("")$
       print("")$
       print("marginal cost of reducing principal balance")$
       print("")$
       print("MC(B) = ",MC(B,L),"")$
       print("")$
```

marginal benefit of reducing principal balance

$$MB(B) = \frac{\mathrm{p}\left(\frac{B}{V}\right)\left(-V+L+B\right)}{V}$$

marginal cost of reducing principal balance

$$MC(B) = 1 - \mathrm{P}\left(\frac{B}{V}\right)$$

MC and MB are reversed because the problem is cast in terms of the costs and benefits of _reducing_ the principal balance.

To plot the MB and MC, we need a distribution. We'll use lognormal.

(%i38)
```
/*  lognormal distribution  */
P(x):=(1/2)*(1+erf((beta*log(x)-mu)/sqrt(2*s^2)))$
p(x):=''(diff(P(x),x))$

/* parameters */
s:0.39$ mu:0.5$ beta:1$

print("")$
print("cumulative lognormal density, when: s=",s,", mu=",mu,", beta=",b
wxplot2d(P,[x,0.01,5],[ylabel,"probability"])$
print("")$
PAtOne:''(float(P(1)))$  PAtOne:''(float(round(10000*PAtOne)/10000))$
PAtTwo:''(float(P(2)))$  PAtTwo:''(float(round(10000*PAtTwo)/10000))$
print("when x=1, probability is:",PAtOne)$
print("when x=2, probability is:",PAtTwo)$
print("")$
```

cumulative lognormal density, when: s= 0.39 *, mu=* 0.5 *, beta=* 1

(%t45)



when x=1, probability is: 0.0999
when x=2, probability is: 0.6898

Let's plot E[port] as a function of the principal balance. Then let's plot the Marginal Benefit and Marginal Cost of principal balance reductions.

For simplicity, we will normalize the home's sale value to one, so that we can quickly calculate the optimal balance as a percentage of the home's value.

Then we'll do comparative static exercises with "L." As a practical matter, "L" would increase if government made foreclosure more difficult.

(%i54)
```
/* normalize the sale value to one and then refresh functions */
/* to incorporate the values and probability functions above  */
V:1$    EPort(B,L):=''(EPort(B,L))$
MB(B,L):=''(MB(B,L))$  MC(B,L):=''(MC(B,L))$
```

(%i58)
```
print("")$
wxplot2d(EPort(B,0.05),[B,1,3],
    [xlabel,"principal balance"],[ylabel,"expected value"])$
print("")$
```

(%t59)



```
(%i61) print("")$
       wxplot2d([MC(B,L),MB(B,0.05)],[B,1,3],[legend,"MC","MB"],
           [xlabel,"principal balance"],[ylabel,"MC and MB"])$
       print("")$
```

(%t62)



MC and MB are reversed because the problem is cast in terms of the costs and benefits of _reducing_ the principal balance.

Now, let's calculate the optimal principal balance and do some comparative statics exercises.

(%i64)
```
/* comparative statics exercises */

/* set L to 5 percent of value and calculate B at maximum */
maxLA:subst(lbfgs(-EPort(B,0.05),[B],[1.01],0.0001,[-1,0]),B)$
maxLA:''(float(round(10000*maxLA)/10000))$

/* set L to 20 percent of value and calculate B at maximum */
maxLB:subst(lbfgs(-EPort(B,0.20),[B],[1.01],0.0001,[-1,0]),B)$
maxLB:''(float(round(10000*maxLB)/10000))$

print("")$
print("when L = 0.05, the optimal B is: ",maxLA)$
print("when L = 0.20, the optimal B is: ",maxLB)$
print("")$
print("In both cases, the optimal principal balance exceeds sale value"
print("(recall that: V=1). In other words, it's in the lender's interes
print("to let the borrower go \"underwater\" ... just not too deep.")$
print("")$
print("But notice that optimal B is lower when L is higher.")$
print("Higher legal fees reduce the optimal principal balance.")$
print("")$
```

when L = 0.05, the optimal B is: 1.7218

when L = 0.20, the optimal B is: 1.6208

In both cases, the optimal principal balance exceeds sale value
(recall that: V=1). In other words, it's in the lender's interest
to let the borrower go "underwater" ... just not too deep.

But notice that optimal B is lower when L is higher.
Higher legal fees reduce the optimal principal balance.

```
(%i79) print("")$
print("Higher legal fees shift the MB curve upward, reducing optimal B
print("")$
wxplot2d([MC(B,L),MB(B,0.05),MB(B,0.20)],[B,1,3],
    [xlabel,"principal balance"],[ylabel,"MB and MC"],
    [legend,"MC","MB(L=0.05)","MB(L=0.20)"])$
print("")$
```

Higher legal fees shift the MB curve upward, reducing optimal B.

(%t82)